
October 6, 2021 

 

While studying some of the minutia of telephone switching systems, I ran across a 

brief article on the use of MF (multi-frequency) tone signaling on trunk lines between 

telephone exchanges.  These are the very same frequencies used in the legendary 

Blue Box encoders to re-route long distance calls after seizing the trunk with a 2600 

Hz tone and forcing a disconnect on the distant end. 

 

In the normal operation of a long-distance call, the originating telephone exchange 

seizes a trunk line by going off-hook and waits for the destination office to “wink” 

(going momentarily off-hook) to signal that it’s ready to accept the destination 

telephone number.  The originating switch then sends the “KP” tone to begin the tone 

sequence.  It then sends the destination phone number using MF tone signaling. 

Although the principle is the same as DTMF, the tones are actually different and 

predate the use of DTMF by more than a decade.  Finally, to signal the end of the 

destination phone number, the originating switch sends the “START” tone and the 

destination switch begin processing the call. 

 

Before direct distance dialing was common, the long-distance operator in the 

originating office would manually key the tones needed to complete the call in a semi-

automatic fashion.  MF tone signaling was a common right up until the end of 

electromechanical switching systems and could often be heard immediately after 

dialing a long-distance call but before the destination number began ringing. 

 

I didn’t have a blue-box handy when I was writing this, so I wrote a quick-and-dirty 

BASIC program for the TI 99/4A to generate the tones.  I chose the 99/4A because it’s 

a classic microcomputer and accepts discrete frequencies as arguments to the “CALL 

SOUND” subprogram. 

 

 
Figure 1 - TI 99/4A Screenshot of the MFTONE Dialer 

Source code for the MFTONE generator program follows.  This was written in plain-

vanilla BASIC for the original TI 99/4A computer without expansion. 

 

 
100 REM  MICHEAL MCCABE  

110 REM  RETROCHALLENGE 2021  

120 REM  MF TONE GENERATOR  

130 CALL CLEAR 

140 DIM F(12,3) 

150 RESTORE 

160 DATA 1300,1500,55 

170 DATA 700,900,55 

180 DATA 700,1100,55 

190 DATA 900,1100,55 

200 DATA 700,1300,55 

210 DATA 900,1300,55 

220 DATA 1100,1300,55 

230 DATA 700,1500,55 

240 DATA 900,1500,55 

250 DATA 1100,1500,55 

260 DATA 1100,1700,110 

270 DATA 1500,1700,55 

280 FOR I=0 TO 11 

290 READ F(I,1),F(I,2),F(I,3) 

300 NEXT I 

310 PRINT "MF TONE GENERATOR FOR" 

320 PRINT "TRUNK DIALING" 

330 PRINT 

340 PRINT "ENTER A STRING OF DIGITS" 

350 PRINT "FOR THE DESTINATION PHONE" 

360 PRINT "NUMBER:" 

370 PRINT 

380 INPUT D$ 

390 PRINT 

400 PRINT "THANK YOU." 

410 PRINT 

414 REM  SEND KP TONE  

415 PRINT "SENDING KP TONE:" 

416 CALL SOUND(F(10,3),F(10,1),7,F(10,2),7) 

417 PRINT 

418 PRINT "SENDING DIGITS..." 

420 FOR I=1 TO LEN(D$) 

430 LET D=VAL(SEG$(D$,I,1)) 

440 CALL SOUND(F(D,3),F(D,1),7,F(D,2),7) 

450 NEXT I 

460 PRINT 

470 PRINT "SENDING ST TONE:" 

480 CALL SOUND(F(11,3),F(11,1),7,F(11,2),7) 

490 PRINT 

500 PRINT "TONE SEQUENCE COMPLETE." 

510 PRINT 

520 INPUT "RUN AGAIN?":Q$ 

530 IF (Q$="Y")+(Q$="YES")THEN 330 

540 END 

 

 

-- Paleoferrosaurus 


